
OVERVIEW

 

Since most detectors are analog type, an analog to dig-
ital converter (ADC) is needed to measure its data by
the digital device. ADC discussed in this application
note is designed to measure slowly changing voltage in
the range from 1.4 to 2.55 Volts with 0.01 Volts accu-
racy (if table is changed, other ADC ranges are possi-
ble using the PIC12C508) and outputs data through it’s
four digital outputs. Input is used to activate ADC mea-
sure (active low) and can be activated through a button,
or a switch, or may be used as a digital line in terminal.
With a few components added it could measure volt-
age, temperature, volume or any other condition with a
proper analog detector. With a simple amplifier added
ADC may be used as a peak meter in a sound system.
And, of course, ADC may be used as a digital compar-
ator to keep some parameters in certain limits through
ADC’s digital outputs.

 

FIGURE 3: SCHEMATIC DIAGRAM 

THE ADC PARTS LIST

 

Measuring mechanism is shown in Figure 2.

 

FIGURE 4: MEASURING MECHANISM

 

First, the  PIC12C508 pin connected to Line In is set to
be output and is driven low to discharge the capacitor.
Then this pin becomes input and the counter starts.
The counter increases while the Line In pin is low. The
pin changes it’s state from low to high when the pin volt-
age is about 1.25 Volts (this voltage strongly depends
on the PICmicro™ supply voltage (5 volts), the leakage
current and the chip itself.  When the pin is raised high,
the counter stops. I assume that the capacitor dis-
charge voltage difference is almost zero (if not then the
pin state change may be assumed slightly smaller).
Since there wouldn’t be two identical PICmicros, I use
the voltage of 1.25 volts as average with minimal volt-
age of about 1.15 volts and the maximum of 1.35 volts.
The variable resistor is used to adjust the timing for volt-
age pin state change and the capacitor difference.

The formula I used to calculate the time when the state
of the pin changes is:

Capacitors: C1 - 0.1

 

µ

 

F

C2 - 47

 

µ

 

f electrolytic

Diodes: D1 – D4 – Any type red light emitting diodes

D5 – Zener diode 4.7 Volts (1N4734A) It is needed to measure voltages 
higher than 5.0 Volts. If the voltage will not exceed 5.0 Volts it is better 
to remove this diode. 

Resistors: R1 – 1.0 MOhm

R2 – 1.0 MOhm variable (adjustment)

R3 –  30 kOhm variable (fine adjustment)

Miscellaneous: U1 – PIC12C508 programmed with ADC code

S1 – Normally open push-button switches

 

U = The voltage when the pin changes its state

E = Input voltage (1.4 to 2.55 volts)

t = Time (t<T)

I used this formula to build a table. But first I have to cal-
culate the resistance for a particular capacitor for the
highest counter (lowest voltage equals 1.395 Volts in
my case). This is needed to fully utilize the table format.

 

R = -t / (C l

 

n

 

(1 - U/E)) = 1.477 Mohms

 

(with a capacitor of 0.1 

 

µ

 

F

where 

When the circuit is ready, connect GP3 to ground and
turn ADC on. Supply Line In terminal with 2.00 Volts
and adjust the circuit until D3 lights up (Be patient while
doing this!). LEDs will light up on the following voltage:

 

QUICK CODE IDEAS

 

I wrote the measuring algorithm as a macro because
there can be more than one analog input in microcon-
troller. Another  macro to compare two 16-bit values.
With some improvement, these values may be 24 or 32
bits wide. Depending on the results, it branches to
three different places.

 

MICROCHIP TOOLS USED

 

This code was written and debugged with MPLAB
for Windows/16 Version 3.22.02

Assembler/Compiler version: MPASM v01.50
using P12C508.inc

t =
 

MHC (5 MLC+2) uS

MHC = measure high counter byte 

MLC =  measure low counter byte

D1 = 1.39 Volts or less

D2 = 1.50 Volts

D3 = 2.00 Volts

D4 = 2.56 Volts or higher

APPENDIX B: SOURCE CODE

 

;ADC
;Author: Kirill Yelizarov

LIST P=PIC12C508, R=DEC
INCLUDE <p12c508.inc>

__CONFIG _IntRC_OSC & _WDT_OFF & _CP_OFF & _MCLRE_OFF

V_Low equ 0 ;Low voltage indicator
V_1_5 equ 1 ;1.5 Volts indicator
V_2_0 equ 2 ;2.0 Volts indicator
Button equ 3 ;Enable measure button
V_High equ 4 ;High voltage indicator
LineIn equ 5 ;ADC line in GPIO pin

TrisReg equ 0x07 ;Tris register
MLC equ 0x08 ;Measure low time
MHC equ 0x09 ;Measure high time
TLC equ 0x0a ;Table low time
THC equ 0x0b ;Table high time
Voltage equ 0x0c ;Voltage on line in
Value equ 0x0d ;Multiporpose register

;                      ------------ M A C R O ----------

;This macro will perform a measuring method described in the document
Measure macro reg,pin,trisr,dh,dl
;reg - register (GPIO for 12C508)
;pin - microcontroller's pin assigned for line in
;trisr - Tris register local data
;dh - counter high byte
;dl - counter low byte

local next
local nextm
local out

bcf trisr,pin ;set line in for output
movf trisr,W
tris reg

bcf reg,pin ;set line in pin low

clrf dh ;wait 1 ms
next:

nop
decfsz dh,F
goto next

clrf dh ;Reset counter
clrf dl

bsf trisr,pin ;set line in for input
movf trisr,W
tris reg

nextm:
btfsc reg,pin ;time=MHC(5MLC+2) us
goto out
incfsz dl,F
goto nextm
incfsz dh,F
goto nextm
comf dh,F ;An overflow occures

comf dl,F ;Voltage is too low if got here
out:

endm

;This macro can compare two 16 bit digits and it do not change them
;Uses W register and flags
;If dh(high),dl(low) is smaller than th(high),tl(low) it branches to slab
;If dh,dl is larger than th,tl it branches to llab
;If dh,dl is equal to th,tl it goes to the end of macro
Compare macro th,tl,dh,dl,slab,llab

movf th,W
subwf dh,W
btfss STATUS,C
goto slab ;if result is negative then go to SMALLER label
btfss STATUS,Z
goto llab ;if not zero then go to LARGER label
movf tl,W
subwf dl,W
btfss STATUS,C
goto slab ;if result is negative then go to SMALLER label
btfss STATUS,Z
goto llab ;if not zero then go to LARGER label

;if got here they are equal
endm

;                      ------------ C O D E ------------

org 0x00
goto Start

VoltageTable
movwf PCL

;Voltage too low
retlw 255
retlw 136

;1.40
retlw 248
retlw 207

;1.41
retlw 242
retlw 142

;1.42
retlw 236
retlw 181

;1.43
retlw 231
retlw 57

;1.44
retlw 226
retlw 17

;1.45
retlw 221
retlw 51

;1.46
retlw 216
retlw 153

;1.47
retlw 212
retlw 59

;1.48
retlw 208
retlw 22

;1.49
retlw 204
retlw 36

;1.50
retlw 200
retlw 97

;1.51
retlw 196
retlw 201

;1.52
retlw 193
retlw 88

;1.53
retlw 190
retlw 13

;1.54
retlw 186
retlw 229

;1.55
retlw 183
retlw 220

;1.56
retlw 180
retlw 241

;1.57
retlw 178
retlw 33

;1.58
retlw 175
retlw 109

;1.59
retlw 172
retlw 209

;1.60
retlw 170
retlw 76

;1.61
retlw 167
retlw 221

;1.62
retlw 165
retlw 130

;1.63
retlw 163
retlw 59

;1.64
retlw 161
retlw 6

;1.65
retlw 158
retlw 227

;1.66
retlw 156
retlw 208

;1.67
retlw 154
retlw 205

;1.68
retlw 152
retlw 217

;1.69
retlw 150
retlw 243

;1.70
retlw 149
retlw 27

;1.71
retlw 147
retlw 80

;1.72
retlw 145
retlw 145

;1.73
retlw 143
retlw 222

;1.74
retlw 142
retlw 54

;1.75
retlw 140
retlw 154

;1.76
retlw 139
retlw 7

;1.77
retlw 137
retlw 127

;1.78
retlw 135
retlw 255

;1.79
retlw 134
retlw 138

;1.80
retlw 133
retlw 29

;1.81
retlw 131
retlw 185

;1.82
retlw 130
retlw 93

;1.83
retlw 129
retlw 8

;1.84
retlw 127
retlw 188

;1.85
retlw 126
retlw 118

;1.86
retlw 125
retlw 56

;1.87
retlw 124
retlw 0

;1.88
retlw 122
retlw 208

;1.89
retlw 121
retlw 165

;1.90
retlw 120
retlw 128

;1.91
retlw 119
retlw 98

;1.92
retlw 118
retlw 73

;1.93
retlw 117
retlw 54

;1.94
retlw 116
retlw 40

;1.95
retlw 115
retlw 31

;1.96
retlw 114
retlw 28

;1.97
retlw 113
retlw 29

;1.98
retlw 112
retlw 35

;1.99
retlw 111
retlw 46

;2.00
retlw 110
retlw 61

;2.01
retlw 109
retlw 81

;2.02
retlw 108
retlw 105

;2.03
retlw 107
retlw 132

;2.04
retlw 106
retlw 164

;2.05
retlw 105
retlw 200

;2.06
retlw 104
retlw 240

;2.07
retlw 104
retlw 27

;2.08
retlw 103
retlw 74

;2.09
retlw 102
retlw 124

;2.10
retlw 101
retlw 178

;2.11
retlw 100
retlw 235

;2.12
retlw 100
retlw 39

;2.13
retlw 99
retlw 102

;2.14
retlw 98
retlw 168

;2.15
retlw 97
retlw 238

;2.16
retlw 97
retlw 54

;2.17
retlw 96
retlw 129

;2.18
retlw 95
retlw 207

;2.19
retlw 95
retlw 31

;2.20
retlw 94
retlw 115

;2.21
retlw 93
retlw 200

;2.22
retlw 93
retlw 32

;2.23
retlw 92
retlw 123

;2.24
retlw 91
retlw 217

;2.25
retlw 91
retlw 56

;2.26
retlw 90
retlw 154

;2.27
retlw 89
retlw 254

;2.28
retlw 89
retlw 100

;2.29
retlw 88
retlw 204

;2.30
retlw 88
retlw 55

;2.31
retlw 87
retlw 163

;2.32
retlw 87
retlw 17

;2.33
retlw 86
retlw 130

;2.34
retlw 85
retlw 245

;2.35
retlw 85
retlw 105

;2.36
retlw 84
retlw 223

;2.37
retlw 84
retlw 87

;2.38
retlw 83
retlw 209

;2.39
retlw 83
retlw 76

;2.40
retlw 82
retlw 201

;2.41
retlw 82
retlw 72

;2.42
retlw 81
retlw 200

;2.43
retlw 81
retlw 74

;2.44
retlw 80
retlw 206

;2.45
retlw 80
retlw 83

;2.46
retlw 79
retlw 218

;2.47
retlw 79
retlw 98

;2.48
retlw 78
retlw 236

;2.49
retlw 78
retlw 119

;2.50
retlw 78
retlw 3

;2.51
retlw 77
retlw 145

;2.52
retlw 77
retlw 32

;2.53
retlw 76
retlw 176

;2.54
retlw 76
retlw 66

;2.55
retlw 75
retlw 213

;Voltage too high

Start:
IF Start>0x100
ERROR "ADC Message: Voltage Table too large."
ENDIF

clrf GPIO
comf GPIO,F ;Turn OFF all LEDs
movlw b'10000000' ;Enable weak pull-up on Button pin
option

clrf TrisReg
bsf TrisReg,LineIn ;Set LineIn as analog input
bsf TrisReg,Button ;Set Button as digital input
tris GPIO

NextMeasure:
btfsc GPIO,Button
goto NextMeasure
movlw 0xE0
movwf MHC
clrf MLC

Wait:
incfsz MLC,F
goto Wait
incfsz MHC,F
goto Wait
btfsc GPIO,Button
goto NextMeasure

Measure GPIO,LineIn,TrisReg,MHC,MLC

movlw low VoltageTable
movwf Value
movlw 138
movwf Voltage ;First test will be done to 1.39 Volts

NextVoltage:
incf Voltage,F
btfsc STATUS,Z
goto Overflow
incf Value,F
movf Value,W
call VoltageTable ;get time high count
movwf THC ;save time high count
incf Value,F
movf Value,W
call VoltageTable ;get time low count
movwf TLC ;save time low count
Compare THC,TLC,MHC,MLC,NextVoltage,SendMessage  ;Compare two words

SendMessage:
movlw b'00010111'
movwf GPIO ;Turn LEDs off
movlw 139 ;test low voltage
subwf Voltage,W
btfsc STATUS,Z
bcf GPIO,V_Low
movlw 150 ;test 1.5 Volts
subwf Voltage,W
btfsc STATUS,Z
bcf GPIO,V_1_5
movlw 200 ;test 2.0 Volts
subwf Voltage,W
btfsc STATUS,Z
bcf GPIO,V_2_0
goto NextMeasure

Overflow:
movlw b'00010111'
movwf GPIO ;Turn LEDs off
bcf GPIO,V_High
goto NextMeasure

org 0x1ff
movlw b'01110000' ;set OSCCAL
end

NOTES:


